

## Mootoriks on 5¼“ Floppiseadme mooter KP4M4-001

```

;*****      AT90S1200      *****
; **  Test, mis kontrollis sammootorit      **
; **    Kui SW1-le vajutada SIIS            **
; **      Moorot pöörleb kord päripäeva või vaastupäeva, **
; **          aeglaselt või kiirelt        **
; **                                          **
; ** Fail      : "SAMM_KII.asm"             **
; ** Kuupäev   : 01.01.2004                 **
; ** Autor     : Jüri Puhang                **
; ** Protsesor : AT90S1200                 **
; **                                          **
; ** SELGITUS:                             **
; ** Sammootori mähised on programmaatori STK500 **
; ** läbi draiverite pordile B ühendatud.   **
; ** Programm skaneerib lüliteid ja kui on vajutatud, **
; ** siis suunatakse ühe sammu sooritamiseks programmi, **
; ** mis teeb sammu ja viivise peale seda.   **
;*****
;**** Initialization ****

    ldi    r16,$00      ; r16      on tööregister
                          ; ajutise info jaoks.
    out    $11,r16     ; PortD ($10) on sisend-
                          ; register SW-d.
                          ; Suuna määrab DDRD ($11)
                          ;   DDD0 - DDD7 := 00000000
                          ;   Kõik on sisendid.
    ldi    r16,0xff    ; r16      tööregister
    out    0x17,r16    ; PortB ($18) on väljund-
                          ; register millel on mootori
                          ; mähis.
                          ; Suund määrab DDRB ($17)
                          ;   DDB0 - DDB7 := 11111111
                          ;   Kõik on väljundid.
;   NB!      2-nd arv 1101 0111 võib AVR Assembleri
;   jaoks kirjutada kujul kas $d7 või 0xd7

;-----
; Kui lülitile on vajutatud,
; on vastav sisend "0" ja on ei ole,
; siis on vastav sisend "1".
;-----
; Sammootori mähis on ühendatud nii:
; -mähis 1a Port PB0 (Kontekt 1)
; -mähis 1b Port PB2 (Kontekt 3)
; -mähis 2a Port PB1 (Kontekt 2)
; -mähis 2b Port PB3 (Kontekt 4)
; Kui on vastav väljund "1" siis läbib
; mootori mähist vool.
;-----

```



## Mootoriks on 5¼“ Floppiseadme mooter KP4M4-001

```

    ldi    r20,1      ; Vaikimisi aeglane kiirus
    ldi    r16,0      ; ja mootori mähised vooluta.
LOOP: out  0x18,r16   ; Välja mootorile.
JARGM:  sbis  0x10,0; Kui PortD,      pin0 == 0 (SW0 sees)
    ldi    r16,0x00   ; SIIS STOPP (Mähised vooluta)
    sbis  0x10,1      ; Kui PortD,      pin1 == 0 (SW1 sees)
    rjmp  P_AEGL      ; SIIS Päripäeva aeglaselt
    sbis  0x10,2      ; Kui PortD,      pin2 == 0 (SW2 sees)
    rjmp  V_AEGL      ; SIIS Vastupäeva aeglaselt
    sbis  0x10,3      ; Kui PortD,      pin3 == 0 (SW3 sees)
    rjmp  P_KIIRE     ; SIIS Päripäeva kiirelt
    sbis  0x10,4      ; Kui PortD,      pin4 == 0 (SW4 sees)
    rjmp  V_KIIRE     ; SIIS Vastupäeva kiirelt
    rjmp  LOOP        ;

;*****
;**** Mootor pöörleb päripäeva *****
;*****
P_KIIRE:
    ldi    r20,1      ; Välist tsüklit ei ole
    rjmp  PARIPAEVA  ;
P_AEGL:  ldi    r20,11; 10 välist tsüklit
PARIPAEVA:
    rol   r16         ; Nihe vasakule = PÄRIPÄEVA
    andi  r16,0x0F    ; Kontroll, kas mähis on vooluta?
    brne  VIIVIS     ; Kui on, siis 1a mähis voolu
    ldi   r16,0x01    ; alla. ( PBO := 1 )
    rjmp  VIIVIS     ;

;*****
;**** Mootor pöörleb vastupäeva *****
;*****
V_KIIRE:ldi r20,1 ; Välist tsüklit ei ole
    rjmp  VASTUPAEVA;
V_AEGL:  ldi    r20,11      ; 10 välist tsüklit
VASTUPAEVA:
    ror   r16         ; Nihe paremale = VASTUPÄEVA
    andi  r16,0x0F    ; Kontroll, kas mähis on vooluta?
    brne  VIIVIS     ; Kui on, siis 2b mähis voolu
    ldi   r16,0x08    ; alla. ( PB3 := 1 )

;*****
;**** Ootamine, enne kui väljastada LED-idele *****
;*****
VIIVIS:
    out  0x18,r16 ; Uus samm tehakse enne viivist
    mov  r19,r20 ; r20 on aeglane või kiire
DLYA:  ldi  r18,60 ; KIIRUSE MÄÄRAMISE
    ;-----
    ; Konstant 250 180 120 60
    ; Täispööre
    ; sek.KIIRE 11 s 8 s 5 s 2,5 s
    ; sek.AEGLANE 115s 80 s 50 s 27 s
    ;-----
DLY:  dec  r17      ; kiirus.
    brne DLY      ; Viivis 256 x 256 tsüklit
    dec  r18
    brne DLY
    dec  r19
    brne DLYA
    rjmp JARGM
;*****

```